
1
Registration
1.1 Split of f_IMS_A_1_1_MessageHeader_CommonChecks
	  function f_IMS_Register_MessageHeader_CommonChecks(MessageHeader p_MessageHeader,

                                                     InternetProtocol_Type p_Protocol,

                                                     boolean p_CheckAuthorization := true) runs on IMS_PTC

  {

    f_IMS_MessageHeader_Request_CheckVia(p_MessageHeader, p_Protocol);  // @sic R5s120727 change 7, 8 sic@

    f_IMS_MessageHeader_CheckContactAddr(p_MessageHeader);      // @sic R5s120727 change 7, 8 sic@

    // Check 'To'

    f_IMS_MessageHeader_CheckTo(p_MessageHeader, noToTag);

    // Check 'SecurityClient'

    if (ispresent(p_MessageHeader.securityClient)) {                               /* 'SecurityClient' is only included for A1, A2 but not for A3 */

      fl_IMS_SecurityClient_Check(p_MessageHeader.securityClient);

    }

    // Check 'Authorization'

    if (p_CheckAuthorization and ispresent(p_MessageHeader.authorization)) {       /* 'Authorization' is only included for A2 but optional for A1, A3 */

      fl_IMS_Authorization_Check(p_MessageHeader.authorization);

    }

  }

	  function f_IMS_A_1_1_MessageHeader_CommonChecks(MessageHeader p_MessageHeader,

                                                  InternetProtocol_Type p_Protocol,

                                                  boolean p_CondA7 := false,

                                                  template (omit) IP_AddrInfo_Type p_UE_Address := omit,

                                                  template (omit) charstring p_ExplicitExpiry := omit,

                                                  boolean p_CheckAuthorization := true) runs on IMS_PTC

  {

    var ContactAddress v_ContactAddress;

    var boolean v_IsDeregistration := ispresent(p_ExplicitExpiry) and (str2int(valueof(p_ExplicitExpiry)) == 0);

    f_IMS_Register_MessageHeader_CommonChecks(p_MessageHeader, p_Protocol, p_CheckAuthorization);




    // Check Contact header

    if (ispresent(p_MessageHeader.contact.contactBody.contactAddresses[0])) {     /* NOTE: is not be present when 'contact' is wildcard e.g. in case of de-registration

                                                                                   *       but in this case the respective receive template needs to allow 'cr_ContactWildcard' */

      v_ContactAddress := p_MessageHeader.contact.contactBody.contactAddresses[0];

      if (pc_IMS_GRUUsInSIP) {                                                    // check 'c-p-instance'

        // @sic R5-153766: Check of instance-id shall be done for GRUU too sic@

        if (not match(v_ContactAddress.contactParams, cr_CPInstanceInContactParams)) {

          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid instance id in Contact header");

        }

      }

      if (not v_IsDeregistration and not f_GetTestcaseAttrib_IMS_UTRAN (testcasename())) {             /* @sic R5-142956: in case of deregistration no further contact parameters need to be checked, R5-169130 or if UTRAN_IMS Testsuite sic@ */

        fl_IMS_Register_ContactHeader_CheckParams(v_ContactAddress, p_CondA7);

      }

      if (isvalue(p_UE_Address)) {

        fl_IMS_ContactAddress_CheckHostname(v_ContactAddress, valueof(p_UE_Address));

      }

    }









    fl_IMS_MessageHeader_CheckExpires(p_MessageHeader, p_ExplicitExpiry);

  }


1.2
Split of f_IMS_REGISTER_InitialRequest
	  function f_IMS_REGISTER_InitialRequestCommon(IMS_IpcanPortType_Type p_IpcanType := IPCAN_E,

                                               template (omit) IMS_DATA_REQ p_IMS_DATA_REQ := omit,

                                               boolean p_SigCompIsActive := false) runs on IMS_PTC return IMS_DATA_REQ

  {

    var IPCAN_INFO_Type v_IpcanInfo;

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var IMS_RoutingInfo_Type v_RoutingInfo_UL;

    var InternetProtocol_Type v_Protocol;

    var IP_AddrInfo_Type v_UE_Address;

    var IP_AddrInfo_Type v_NW_Address;

    var MessageHeader v_MessageHeader;

    var template (present) integer v_CseqValue := ?;

    var boolean v_Match := false;

    v_IMS_DATA_REQ := fl_IMS_ReceiveOrGetRegisterRequest(p_IMS_DATA_REQ, v_CseqValue);   /* @sic R5-176758: v_CseqValue sic@ */

    v_MessageHeader := v_IMS_DATA_REQ.Request.Register.msgHeader;

    v_Protocol := v_IMS_DATA_REQ.RoutingInfo.Protocol;

    select (p_IpcanType) {

      case (IPCAN_E) {

        v_IpcanInfo := f_IMS_IPCAN_Query(IPCAN);

      }

      case else {

        v_IpcanInfo := f_IMS_IPCAN_Query(OtherIPCAN);

      }

    }

    f_IMS_PTC_SetIpcanInfo(v_IpcanInfo);

    v_Match := match(v_MessageHeader, f_IMS_RegisterRequest_MessageHeaderRX_Initial(v_Protocol, v_CseqValue));         /* @sic R5s130369 sic@

                                                                                                                          @sic R5-176758: v_CseqValue sic@ */

    if (not v_Match) {

      f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "invalid REGISTER Request");

    }

    v_RoutingInfo_UL := v_IMS_DATA_REQ.RoutingInfo;

    v_UE_Address := v_RoutingInfo_UL.UE_Address;

    v_NW_Address := v_RoutingInfo_UL.NW_Address;

    v_Protocol := v_RoutingInfo_UL.Protocol;

    f_IMS_PTC_ImsInfo_Init(v_UE_Address, v_NW_Address, v_IMS_DATA_REQ.Request.Register, p_SigCompIsActive);

    return v_IMS_DATA_REQ;

  }

	  function f_IMS_REGISTER_InitialRequest(IMS_IpcanPortType_Type p_IpcanType := IPCAN_E,

                                         template (omit) IMS_DATA_REQ p_IMS_DATA_REQ := omit,

                                         template (omit) charstring p_ExplicitExpiry := omit,

                                         boolean p_SigCompIsActive := false,

                                         boolean p_CondA7 := false,

                                         boolean p_CondA3 := false,

                                         boolean p_CondA14 := false) runs on IMS_PTC return IMS_DATA_REQ

  { /* p_CondA7  .. condition A7 acc. to 34.229-1 Annex A.1.1 (Initial IMS emergency registration)

       p_CondA3  .. REGISTER for GIBA; @sic R5s140356: p_CondA3 sic@

       p_CondA14 .. REGISTER for FBBA; @sic R5-160290 FBBA registration: p_CondA14 sic@  */

    /* @sic R5s130756 additional changes: p_CondA3 removed sic@ */

    /* @sic R5-134659: check of Route header removed sic@ */

    /* @sic R5s160133: p_IpcanType sic@ */


    var IMS_DATA_REQ v_IMS_DATA_REQ;


    var InternetProtocol_Type v_Protocol;

    var IP_AddrInfo_Type v_UE_Address;


    var MessageHeader v_MessageHeader;




    v_IMS_DATA_REQ := f_IMS_REGISTER_InitialRequestCommon(p_IpcanType, p_IMS_DATA_REQ, p_SigCompIsActive);
    v_MessageHeader := v_IMS_DATA_REQ.Request.Register.msgHeader;

    v_Protocol := v_IMS_DATA_REQ.RoutingInfo.Protocol; 
    v_UE_Address := v_IMS_DATA_REQ.RoutingInfo.UE_Address;



















    // additional checks

    f_IMS_A_1_1_MessageHeader_CommonChecks(v_MessageHeader, v_Protocol, p_CondA7, v_UE_Address, p_ExplicitExpiry);

    f_IMS_A_1_1_MessageHeader_CheckPublicUserIdentity(v_MessageHeader, p_CondA7, p_CondA3, p_CondA14);  // check 'To' and 'From' field (needed for initial registration only); @sic R5s140356: p_CondA3 sic@

    if (p_SigCompIsActive) {

      f_IMS_MessageHeader_CheckSigComp(v_MessageHeader);

    }

    return v_IMS_DATA_REQ;

  }


1.3
Split of f_IMS_REGISTER_SubsequentRequest
	  function f_IMS_REGISTER_SubsequentRequestCommon(template (omit) IMS_DATA_REQ p_IMS_DATA_REQ := omit,

                                                  integer p_NonceCount := 1) runs on IMS_PTC return IMS_DATA_REQ

  {

    var integer v_CseqValue := f_IMS_PTC_ImsInfo_CseqIncr(register);                     /* @sic R5-176758 sic@ */

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var REGISTER_Request v_RegisterReq;

    var MessageHeader v_MessageHeader;

    var InternetProtocol_Type v_Protocol;

    var boolean v_Match := false;

    // Receive/Get second REGISTER on secure port:

    v_IMS_DATA_REQ := fl_IMS_ReceiveOrGetRegisterRequest(p_IMS_DATA_REQ, v_CseqValue);   /* @sic R5-176758: v_CseqValue, to distinguish subsequent REGISTER from retransmission of initial REGISTER sic@ */

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    v_Protocol := v_IMS_DATA_REQ.RoutingInfo.Protocol;

    v_MessageHeader := v_RegisterReq.msgHeader;

    v_Match := match(v_MessageHeader, f_IMS_RegisterRequest_MessageHeaderRX_Subsequent(p_NonceCount, v_Protocol, v_CseqValue));   /* @sic R5-176758: v_CseqValue sic@ */

    if (not v_Match) {

      f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "invalid REGISTER Request");

    }

    f_IMS_PTC_ImsInfo_SetContactUrl(f_MessageHeader_GetContactSipUrl(v_MessageHeader));

    return v_IMS_DATA_REQ;

  }

	  function f_IMS_REGISTER_SubsequentRequest(template (omit) IMS_DATA_REQ p_IMS_DATA_REQ := omit,

                                            boolean p_CondA7 := false,

                                            integer p_NonceCount := 1,

                                            template (omit) charstring p_ExplicitExpiry := omit) runs on IMS_PTC  return IMS_DATA_REQ

  { /* p_CondA7  .. condition A7 acc. to 34.229-1 Annex A.1.1 (Initial IMS emergency registration)

     * p_NonceCount  ..  Acc. to A.1.1 "counter to indicate how many times UE has sent the same value of nonce within successive REGISTERs, initial value shall be 1"

     *                   => during initial registration "1" is used per default; for each re-authentication (=> new nonce) the nonce-counter is reset to "1"

     */

    var IP_AddrInfo_Type v_UE_Address := f_IMS_PTC_UE_Address_Get();


    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var REGISTER_Request v_RegisterReq;

    var MessageHeader v_MessageHeader;

    var InternetProtocol_Type v_Protocol;


    // Receive/Get second REGISTER on secure port:

    v_IMS_DATA_REQ := f_IMS_REGISTER_SubsequentRequestCommon(p_IMS_DATA_REQ, p_NonceCount);


    v_Protocol := v_IMS_DATA_REQ.RoutingInfo.Protocol; 
    v_MessageHeader := v_IMS_DATA_REQ.Request.Register.msgHeader;






      f_IMS_A_1_1_MessageHeader_CommonChecks(v_MessageHeader, v_Protocol, p_CondA7, v_UE_Address, p_ExplicitExpiry); // @sic R5s120945 sic@

      if (f_IMS_PTC_ImsInfo_SigCompStarted()) {

        f_IMS_MessageHeader_CheckSigComp(v_MessageHeader);

      }



    return v_IMS_DATA_REQ;

  }


1.4
Split of f_IMS_PTC_Deregistration
	  function f_IMS_REGISTER_DeregistrationRequest(template (omit) IMS_DATA_REQ p_IMS_DATA_REQ := omit) runs on IMS_PTC return IMS_DATA_REQ

  {

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var MessageHeader v_MessageHeader;

    var integer v_CseqValue := f_IMS_PTC_ImsInfo_CseqIncr(register);

    var boolean v_Match := false;

    v_IMS_DATA_REQ := fl_IMS_ReceiveOrGetRegisterRequest(p_IMS_DATA_REQ, v_CseqValue);   /* @sic R5-176758: v_CseqValue sic@ */

    v_MessageHeader := v_IMS_DATA_REQ.Request.Register.msgHeader;

    v_Match := match(v_MessageHeader, f_IMS_RegisterRequest_MessageHeaderRX_Derregister(v_CseqValue));   /* @sic R5-176758: v_CseqValue sic@ */

    if (not v_Match) {

      f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "invalid REGISTER Request for de-registration");

    }

    return v_IMS_DATA_REQ;

  }

	  function f_IMS_PTC_Deregistration(template (omit) IMS_DATA_REQ p_IMS_DATA_REQ := omit) runs on IMS_PTC

  { /* @sic R5-140902: no delay needed anymore to wait for TCP close sic@ */

    /* @sic R5s170445: closing of TCP connection done outside this function again sic@ */

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var MessageHeader v_MessageHeader;

    var IP_AddrInfo_Type v_UE_Address := f_IMS_PTC_UE_Address_Get();


    var InternetProtocol_Type v_Protocol;

    var charstring v_Expires := "0";



    v_IMS_DATA_REQ := f_IMS_REGISTER_DeregistrationRequest(p_IMS_DATA_REQ);
    v_MessageHeader := v_IMS_DATA_REQ.Request.Register.msgHeader;

    v_Protocol := v_IMS_DATA_REQ.RoutingInfo.Protocol;






      f_IMS_A_1_1_MessageHeader_CommonChecks(v_MessageHeader, v_Protocol, -, v_UE_Address, v_Expires);   // check expires: to be 0 and present either in Expires or in contact or in both (-> 8.3.4/8.13.4)


    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo),

                                     cs_Response(c_statusLine200, f_IMS_RegisterResponse_200_MessageHeaderTX(v_IMS_DATA_REQ.Request.Register, -, v_Expires))));  // @sic R5s130274: v_Expires sic@

  }


1.5
Split of f_IMS_Register_SecurityInit
	  function f_IMS_Register_SecurityInit_ImsSecurity(REGISTER_Request p_RegisterReq,

                                                   B128_Type p_Rand := tsc_IMS_AuthRAND,

                                                   boolean p_StartSecurity := true) runs on IMS_PTC

  {

    var Common_AuthenticationParams_Type v_AuthenticationParams;

    var SecurityClientParams_Type v_SecurityClientParams;

    var IP_AddrInfo_Type v_UE_Address := f_IMS_PTC_UE_Address_Get();

    var IP_AddrInfo_Type v_NW_Address := f_IMS_PTC_NW_Address_Get();

    var SipUrl v_ContactUrl := f_MessageHeader_GetContactSipUrl(p_RegisterReq.msgHeader);

    var template (omit) PortNumber_Type v_UnprotectedPort_us := f_SIP_ContactUrl_GetPort(v_ContactUrl);           // may be omit in which case 5060 is chosen by the IP PTC

    var template (omit) IMS_SecurityInfo_Type v_SecurityInfo;

    v_AuthenticationParams := f_IMS_AuthenticationInit(p_Rand);

    v_SecurityClientParams := f_IMS_Register_GetSecurityClientParams(p_RegisterReq); // @sic R5s140266 sic@

    /* initialise or re-initialise SPIs and protected ports of the network side: needed e.g. for f_IMS_PTC_BuildSecurityServerParam as used in f_IMS_RegisterResponse_401_MessageHeaderTX */

    f_IMS_PTC_Security_Init(v_AuthenticationParams, v_SecurityClientParams.Port_us, v_SecurityClientParams.Port_uc, v_SecurityClientParams.SPI_us, v_SecurityClientParams.SPI_uc);

    if (p_StartSecurity) {

      f_IMS_InstallIPsecKeys(v_AuthenticationParams.IK, v_AuthenticationParams.CK);

      v_SecurityInfo := cs_IMS_SecurityInfo(f_IMS_PTC_Security_GetProtectedPorts(),

                                            f_IMS_PTC_Security_GetSPIs(),

                                            f_IMS_PTC_Security_GetIntegrityAlgorithm(),

                                            f_IMS_PTC_Security_GetCipheringAlgorithm());

      // @sic R5s170050: f_IMS_PortsAndSecurityConfig moved here sic@

      f_IMS_PortsAndSecurityConfig(v_UnprotectedPort_us,

                                   cs_IMS_RegistrationInfo(v_NW_Address,             // @sic R5s150268 change 2 sic@

                                                           v_UE_Address,

                                                           v_SecurityInfo));

    }

  }

	  function f_IMS_Register_SecurityInit(REGISTER_Request p_RegisterReq,

                                       B128_Type p_Rand := tsc_IMS_AuthRAND,

                                       boolean p_StartSecurity := true) runs on IMS_PTC

  { /* @sic R5s120858 change 6..8 - MCC160 implementation: additional parameter p_Unprotected to deal with GIBA sic@ */

    /* @sic R5s130266 change 1.3, 4, 5 - MCC160 Implementation: SPIs for re-authentication sic@ */

    /* @sic R5s150873 p_StartSecurity instead of p_Unprotected sic@ */

    var Common_AuthenticationParams_Type v_AuthenticationParams;


    var IP_AddrInfo_Type v_UE_Address := f_IMS_PTC_UE_Address_Get();

    var IP_AddrInfo_Type v_NW_Address := f_IMS_PTC_NW_Address_Get();

    var SipUrl v_ContactUrl := f_MessageHeader_GetContactSipUrl(p_RegisterReq.msgHeader);

    var template (omit) PortNumber_Type v_UnprotectedPort_us := f_SIP_ContactUrl_GetPort(v_ContactUrl);           // may be omit in which case 5060 is chosen by the IP PTC


    var charstring v_PasswordFBBA;

    if (ispresent(p_RegisterReq.msgHeader.securityClient)) {                           // @sic R5s150873: false for GIBA sic@

      f_IMS_Register_SecurityInit_ImsSecurity(p_RegisterReq, p_Rand, p_StartSecurity);
















    }

    else {

      if (f_IMS_PTC_SecurityScheme_IsSipDigest()) {

        v_PasswordFBBA := px_DigestPasswordForSIP;  /* @sic R5-170885 sic@ */

        v_AuthenticationParams := f_AuthenticationInit(valueof(cs_CommonAuthParams_Init(p_Rand)));

        f_IMS_PTC_Security_InitFBBA(v_AuthenticationParams, v_PasswordFBBA);

      }

      f_IMS_PortsAndSecurityConfig(v_UnprotectedPort_us, cs_IMS_RegistrationInfo(v_NW_Address, v_UE_Address));

    }

  }


2
Call Control

2.1
Split of f_IMS_BuildSDP_RX
	  function f_IMS_BuildSDP_RX_AnyBW(template (present) SDP_time p_TimeDescription := cr_SDP_Time_0_0,

                                   template (present) SDP_media_field p_SDP_Media_Field,

                                   template (present) SDP_attribute_list p_SDP_MediaAttributes := {},

                                   template (present) SDP_attribute_list p_SDP_PrecondionAttributes := {},

                                   template SDP_bandwidth_list p_Media_Desc_BandwidthList := *) runs on IMS_PTC return template (present) SDP_Message

  { /* @sic R5s150704, R5s150751 - additional changes: new parameters p_MediaBW_AS, p_MediaBW_RS, p_MediaBW_RR sic@ */

    var charstring v_AddrType := f_IMS_PTC_NW_Address_GetTypeStr();

    var template (present) charstring v_IPAddrOrigin := ?;

    var template (present) charstring v_IPAddrConnection := ?;

    var template SDP_attribute_list v_SDP_Attribute_List := f_SDP_Attributes_Concat_RX(p_SDP_MediaAttributes, p_SDP_PrecondionAttributes);

    var template (present) SDP_media_desc v_SDP_Media_Desc;

    var template (present) SDP_media_desc_list v_SDP_Media_Desc_List;

    if (lengthof(v_SDP_Attribute_List) > 0) {

      v_SDP_Attribute_List := superset(all from v_SDP_Attribute_List);

    } else {

      v_SDP_Attribute_List := *;

    }

    v_SDP_Media_Desc := cr_SDP_Media_Desc(p_SDP_Media_Field,

                                          cr_SDP_Connection(v_AddrType, v_IPAddrConnection),

                                          p_Media_Desc_BandwidthList,

                                          v_SDP_Attribute_List);

    v_SDP_Media_Desc_List := { v_SDP_Media_Desc };

    return cr_SDP_Message_Common(v_AddrType, v_IPAddrOrigin, v_IPAddrConnection, -, p_TimeDescription, v_SDP_Media_Desc_List);

  }

	  function f_IMS_BuildSDP_RX(template (present) SDP_time p_TimeDescription := cr_SDP_Time_0_0,

                             template (present) SDP_media_field p_SDP_Media_Field,

                             template (present) SDP_attribute_list p_SDP_MediaAttributes := {},

                             template (present) SDP_attribute_list p_SDP_PrecondionAttributes := {},

                             template (present) integer p_MediaBW_AS := ?,

                             template (present) integer p_MediaBW_RS := ?,

                             template (present) integer p_MediaBW_RR := ?) runs on IMS_PTC return template (present) SDP_Message

  { /* @sic R5s150704, R5s150751 - additional changes: new parameters p_MediaBW_AS, p_MediaBW_RS, p_MediaBW_RR sic@ */





    var template SDP_bandwidth_list v_Media_Desc_BandwidthList := f_SDP_SDP_Media_Desc_BandwidthList(p_SDP_Media_Field, p_MediaBW_AS, p_MediaBW_RS, p_MediaBW_RR);  /* results in * if the stream shall be released (port number == 0)

                                                                                                                                                                       @sic R5s150704, R5s150751 - additional changes sic@ */ 
    var template (present) SDP_Message v_SDP_Message := f_IMS_BuildSDP_RX_AnyBW(p_TimeDescription, p_SDP_Media_Field, p_SDP_MediaAttributes, p_SDP_PrecondionAttributes, v_Media_Desc_BandwidthList);












    return v_SDP_Message;

  }


2.2
Improvement of f_IMS_InviteRequest_ViaTX
NOTE: Change should not have impact on IMS test case as per default for DL requests UDP is used for non-registration test case.
	  function f_IMS_InviteRequest_ViaTX(charstring p_CallerDomain := tsc_IMS_CallerDomain,

                                     integer p_CallerPort := tsc_IMS_CallerPort) runs on IMS_PTC return template (value) Via

  {

    var template (value) ViaBody_List v_ViaBodyList;

    var template (omit) GenericParam v_SigCompParam := f_IMS_PTC_ImsInfo_GetSigCompParam();   // omit when sigcomp is not started

    var charstring v_Host := f_IMS_PTC_Pcscf_Get();

    var integer v_Port_ps := f_IMS_PTC_GetPort_ps();
    var InternetProtocol_Type v_InternetProtocol := f_IMS_PTC_ImsInfo_GetTransportProtocol();

    var charstring v_TransportProtocol1 := f_IMS_ConvertTransportProtocolToString(v_InternetProtocol);
    var charstring v_TransportProtocol2 := "UDP";   /* acc. to via parameter in TS 34.229 A.2.9 udp shall be used

                                                      @sic NOTE: acc. to discussions regarding rejection of R5-130765 at RAN WG5 Meeting #58 UDP is used for all requests in DL sic@ */

    v_ViaBodyList := {

      cs_ViaBody(v_TransportProtocol1, cs_HostPort(v_Host, v_Port_ps),          f_ViaParamsTX(f_IMS_PTC_ImsInfo_GetNextBranch(), v_SigCompParam)),

      cs_ViaBody(v_TransportProtocol2, cs_HostPort("scscf1.3gpp.org"),          f_ViaParamsTX(f_IMS_PTC_ImsInfo_GetNextBranch())),

      cs_ViaBody(v_TransportProtocol2, cs_HostPort("scscf2.3gpp.org"),          f_ViaParamsTX(f_IMS_PTC_ImsInfo_GetNextBranch())),

      cs_ViaBody(v_TransportProtocol2, cs_HostPort("pcscf2.3gpp.org"),          f_ViaParamsTX(f_IMS_PTC_ImsInfo_GetNextBranch())),

      cs_ViaBody(v_TransportProtocol2, cs_HostPort(p_CallerDomain, p_CallerPort), f_ViaParamsTX(f_IMS_PTC_ImsInfo_GetNextBranch()))

    };

    return cs_Via(v_ViaBodyList);

  }


2.3
Split of f_IMS_InviteRequest_MessageHeaderTX
	  function f_IMS_InviteRequest_MessageHeaderTX_Common(OptionTag_List p_OptionTagsForSupported := tsc_OptionTagList_Invite_C11,

                                                      boolean p_IsReInvite := false,

                                                      integer p_CSeqValue := 4711) runs on IMS_PTC return template (value) MessageHeader

  {

    var template (value) MessageHeader v_MessageHeader_Invite := cs_MessageHeader_Dummy;

    var template (value) SipUrl v_PublicUserId_SipUrl := f_IMS_PTC_ImsInfo_GetSipUriTX(defaultId);  // @sic R5-133304: use default public user id sic@

    var charstring v_CallIdString := f_IMS_PTC_GenerateCallId("INVITE");

    var charstring v_FromTag := f_IMS_GenerateTag("INVITE", "From");

    var template (value) RouteBody_List v_RouteBodyList;

    var integer v_CSeqValue;

    var template (value) From v_From;

    var template (value) To v_To;

    var template (value) CallId v_CallId;

    if (not p_IsReInvite) {

      v_CSeqValue := p_CSeqValue;

      v_From   := cs_From(f_SIP_BuildSipUri_TX(tsc_IMS_CallerSipAddrStr), v_FromTag);

      v_To     := cs_ToDef(v_PublicUserId_SipUrl);

      v_CallId := cs_CallId(v_CallIdString);

      v_RouteBodyList := f_IMS_RouteSet_MT_Call_TX();      /* @sic R5s140324 - MCC160 Implementation sic@ */

    }

    else {

      v_CSeqValue := f_IMS_PTC_ImsInfo_CseqIncr(dialogLocal);

      v_From   := f_IMS_RequestInDialog_FromHeaderTX();

      v_To     := f_IMS_RequestInDialog_ToHeaderTX();

      v_CallId := f_IMS_PTC_ImsInfo_DialogGetCallId();

      if (f_IMS_PTC_ImsInfo_DialogIsMO()) {      /* @sic R5s140324 - MCC160 Implementation sic@ */

        v_RouteBodyList := fl_IMS_RouteSet_ReverseOrder(f_IMS_RouteSet_MO_Call_TX());    /* @sic R5-142320, R5-144601 sic@ */

      } else {

        v_RouteBodyList := f_IMS_RouteSet_MT_Call_TX();

      }

    }

    v_MessageHeader_Invite.via             := f_IMS_InviteRequest_ViaTX();  /* @sic R5-134116: New function fl_IMS_InviteRequest_ViaTX sic@ */

    v_MessageHeader_Invite.recordRoute     := cs_RecordRoute(v_RouteBodyList);

    v_MessageHeader_Invite.fromField       := v_From;

    v_MessageHeader_Invite.toField         := v_To;

    v_MessageHeader_Invite.callId          := v_CallId;

    v_MessageHeader_Invite.cSeq            := cs_CseqDef(v_CSeqValue, "INVITE");

    v_MessageHeader_Invite.supported       := cs_Supported(p_OptionTagsForSupported);

    v_MessageHeader_Invite.pCalledPartyID  := cs_PCalledPartyId(v_PublicUserId_SipUrl);

    v_MessageHeader_Invite.maxForwards     := cs_MaxForwardsDef;

    if (not p_IsReInvite) {

      f_IMS_PTC_ImsInfo_DialogInit(dialogLocal, valueof(v_MessageHeader_Invite));

    }

    f_IMS_PTC_ImsInfo_DialogSetRemoteRSeq(omit);       /* reset RSeq for every (re-)INVITE   @sic R5-181190 sic@ */

    return v_MessageHeader_Invite;

  }

	  function f_IMS_InviteRequest_MessageHeaderTX(OptionTag_List p_OptionTagsForSupported := tsc_OptionTagList_Invite_C11,

                                               INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,

                                               boolean p_SrvccAlertingSupported := pc_IMS_SRVCCAlert,

                                               boolean p_MTSI := pc_MultimediaTelephonyService,    /* @sic R5s130510 additional change 1: tsc_IMS_MTSI replaced by pc_MultimediaTelephonyService sic@ */

                                               boolean p_VideoFeatureTag := pc_IMS_Video_FeatureTag,

                                               integer p_CSeqValue := 4711) runs on IMS_PTC return template (value) MessageHeader

  {

    var template (value) MessageHeader v_MessageHeader_Invite;


    var template (omit) SemicolonParam_List v_ContactParams := omit;

    var template (value) GenericParam v_GenericParam_3gpp_icsi_ref;








    select (p_Context) {

      case (A_2_1_A4) {

        v_IsReInvite := false;





      }

      case (A_2_1_A5) {

        v_IsReInvite := true;









      }

      case else {

        FatalError(__FILE__, __LINE__, "unknown context");

      }

    }

    v_MessageHeader_Invite := f_IMS_InviteRequest_MessageHeaderTX_Common(p_OptionTagsForSupported, v_IsReInvite, p_CSeqValue);








    v_MessageHeader_Invite.contentType     := cs_ContentTypeSDP;


    v_MessageHeader_Invite.accept          := cs_Accept({cs_AcceptBody(c_sdpApplication), cs_AcceptBody(c_ims3gppApplication)});

    if (p_MTSI) {

      v_GenericParam_3gpp_icsi_ref := cs_GenericParam_Common(tsc_IMS_FeatureParamId_3gpp_icsi_ref, cs_GenValueQuoted(cs_IMS_EncodedFeatureParamValue_3gpp_icsi_ref));  // @sic R5s130738: string needs to be quoted; R5-142247 sic@

      if (p_Context == A_2_1_A4) {      // @sic R5-142949 sic@

        v_MessageHeader_Invite.pAssertedService := cs_PAssertedService(tsc_IMS_FeatureParamValue_3gpp_icsi_ref);   /* @sic R5s140565 change 1 sic@ */

        v_MessageHeader_Invite.acceptContact    := cs_AcceptContact({v_GenericParam_3gpp_icsi_ref});

      }

      v_ContactParams := f_SemicolonParam_List_Add_TX(v_ContactParams, v_GenericParam_3gpp_icsi_ref);

    }

    if (p_VideoFeatureTag) { // @sic R5s130109 sic@

      v_ContactParams := f_SemicolonParam_List_Add_TX(v_ContactParams, cs_GenericParam("video"));

    }

    v_ContactParams := f_SemicolonParam_List_Add_TX(v_ContactParams, cs_GenericParam("audio"));                     /* @sic R5-165943 sic@ */

    v_MessageHeader_Invite.contact := cs_Contact(f_SIP_BuildSipUri_TX(tsc_IMS_CallerSipAddrStr, tsc_IMS_CallerPort), v_ContactParams);

    if (p_Context != A_2_1_A5) {

      if (p_SrvccAlertingSupported) {   /* @sic R5s140385 change 6 - NOTE: feature caps shall only be inserted when creating a voice or video call sic@ */

        v_MessageHeader_Invite.featureCaps   := cs_FeatureCaps({ cs_GenericParam(tsc_IMS_FeatureCap_SrvccAlerting)});  // @sic R5-145796 sic@

      }


    }


    return v_MessageHeader_Invite;

  }


2.4
Enhancement of f_IMS_ByeRequest_MessageHeaderRX
see change 3.1
	  function f_IMS_ByeRequest_MessageHeaderRX(template (value) INVITE_Request p_InviteRequest,

                                            IMS_CallType_Type p_CallType := NORMAL_CALL,

                                            template (omit) IMS_PTC_DialogIndex_Type p_DialogIndex := omit,

                                            boolean p_IsEmergencyCall := false) runs on IMS_PTC return template (present) MessageHeader

  { /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */

    /* @sic R5s150692 change 19: parameter p_CallType sic@ */

    /* @sic R5-160505: p_A5 := false per default sic@ */

    /* @sic R5-172934: p_A6, p_A5 removed; new parameter p_IsEmergencyCall sic@ */

    /* @sic R5-174544: SessionId removed sic@ */

    var boolean v_A1 := f_IMS_PTC_SecurityScheme_IsImsSecurity();

    var boolean v_A2 := f_IMS_PTC_SecurityScheme_IsGiba();

    var boolean v_A5 := f_IMS_PTC_SecurityScheme_IsSipDigest();

    var boolean v_A6 := p_IsEmergencyCall and not f_IMS_PTC_ImsInfo_IsRegistered();       /* @sic R5-172934 sic@ */

    var boolean v_IsMCX := f_IMS_PTC_TestcaseClass_IsMCX();

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (present) MessageHeader v_MessageHeader_Bye := cr_MessageHeader_Dummy;

    var integer v_CSeqValue := f_IMS_PTC_ImsInfo_CseqIncr(dialogRemote, p_DialogIndex);

    var template (present) RouteBody_List v_RouteBodyList;

    var template (present) Via v_Via;

    var template (present) PAccessNetworkInfo  v_PAccessNetworkInfo := cr_PAccessNetworkInfo(f_IMS_PTC_GetRanType());

    if (not fl_IMS_InviteRequestIsMO(p_InviteRequest)) {                  /* @sic R5-134116: sic@

                                                                             in case of MT call establishment the INVITE sent to the UE contains a Record-Route header with a valid route set */

      v_Via := f_IMS_InviteRequest_ViaRX();

    } else {                                                              /* in case of MO call establishment the route set as for the 183 response shall be used (in reverse order) */

      v_Via := f_IMS_Via_AsInPrevInvite(v_MessageHeader_Invite.via)

    }

    v_RouteBodyList := f_IMS_RouteSet_RouteBodyList_RX(p_CallType, p_DialogIndex);              /* @sic R5s140017 MCC160 implementation: re-INVITE - Route is different for MO and MT calls sic@

                                                                                                   @sic R5s150692 change 19 sic@ */

    v_MessageHeader_Bye.via             := v_Via;

    v_MessageHeader_Bye.route           := cr_Route(v_RouteBodyList);

    v_MessageHeader_Bye.toField         := f_IMS_RequestInDialog_ToHeaderRX(p_DialogIndex);     /* @sic R5s130497 change 1 - MCC160 implementation; R5s140017 MCC160 implementation sic@ */

    v_MessageHeader_Bye.fromField       := f_IMS_RequestInDialog_FromHeaderRX(p_DialogIndex);   /* @sic R5s130497 change 1 - MCC160 implementation; R5s140017 MCC160 implementation sic@ */

    v_MessageHeader_Bye.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Bye.cSeq            := cr_CseqDef(v_CSeqValue, "BYE");

    v_MessageHeader_Bye.maxForwards     := cr_MaxForwardsDef;

    if (v_A6) {       /* @sic R5-172934 sic@ */

      v_MessageHeader_Bye.require            := omit;

      v_MessageHeader_Bye.proxyRequire       := omit;

      v_MessageHeader_Bye.securityVerify     := omit;

    }

    else {

      if (v_A1) {

        v_MessageHeader_Bye.require          := cr_RequireSecAgree;

        v_MessageHeader_Bye.proxyRequire     := cr_ProxyRequireDef;

        if (not v_IsMCX) {

          v_MessageHeader_Bye.securityVerify := f_IMS_PTC_BuildSecurityVerifyHeader();

        } else {

          v_MessageHeader_Bye.securityVerify := omit;              // !!!! not clear in the prose !!!!

        }

      }

      else if (v_A2) {

        v_MessageHeader_Bye.securityVerify   := omit;

      }

      else if (v_A5) {

        v_MessageHeader_Bye.require          := cr_RequireSecAgree;

        v_MessageHeader_Bye.proxyRequire     := cr_ProxyRequireDef;

      }

    }

    if (v_A5) {

      v_MessageHeader_Bye.proxyAuthorization := fl_IMS_Invite_BuildProxyAuthorizationHeader();

    }

    if (v_A1) {

      v_MessageHeader_Bye.pAccessNetworkInfo := v_PAccessNetworkInfo;

    }

    else if (v_A2) {

      v_MessageHeader_Bye.pAccessNetworkInfo := v_PAccessNetworkInfo ifpresent;

    }

    return v_MessageHeader_Bye;

  }


3.
Changes related to Subscribe procedures
3.1
Enhancement of IMS_PTC_DialogIndex_Type
	  type enumerated IMS_PTC_DialogIndex_Type {    /* @status    APPROVED (ENDC, IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, NR5GC, POS) */

    firstDialog,

    secondDialog,

    thirdDialog,

    referDialog,

    confEventDialog,    /* @sic R5-153986: confEventDialog sic@ */

    regEventDialog,     /* @sic R5-153756: needed to unsubscribe (to get the correct CSeq for the NOTIFY) sic@ */

    msgSummaryDialog,    /* @sic R5-153756: needed to unsubscribe (to get the correct CSeq for the NOTIFY) sic@ */
    presenceDialog,     /* Dialog for presence event package, MCX TS 36.579-1 Table 5.5.2.14-1 */

    xcapDiffDialog      /* Dialog for xcap-diff event package, MCX TS 24.379 clause 6.3.5 */
  };

	  const integer tsc_IMS_NoOfDialogs := 8;

	  function fl_IMS_PTC_ImsInfo_GetDialogIndexValue(template (omit) IMS_PTC_DialogIndex_Type p_Index) runs on IMS_PTC return integer

  {

    var integer i;

    if (isvalue(p_Index)) {

      select (valueof(p_Index)) {

        case (firstDialog)      { i := 0; }

        case (secondDialog)     { i := 1; }

        case (thirdDialog,

              referDialog)      { i := 2; }

        case (confEventDialog)  { i := 3; }    /* @sic R5-153986 sic@ */

        case (regEventDialog)   { i := 4; }    /* @sic R5-153756 sic@ */

        case (msgSummaryDialog) { i := 5; }    /* @sic R5-153756, R5s150854 sic@ */ 
        case (presenceDialog)   { i := 6; }

        case (xcapDiffDialog)   { i := 7; }
        case else { FatalError(__FILE__, __LINE__, "invalid dialog"); }

      }

    } else {

      i := vc_IMS_Global.ImsInfo.Dialog.CurrentIndex;

    }

    return i;

  }


3.2
Function f_IMS_SubscribeRequest_MessageHeaderRX
	  function f_IMS_SubscribeRequest_MessageHeaderRX(EventType p_EventType := "reg",

                                                  template SipUrl p_SipUrlTo := omit,

                                                  template (present) charstring p_ExpireTime := "600000")

    runs on IMS_PTC return template (present) MessageHeader

  { /* p_EventType:

     *   "reg"             - normal case acc. A.1.4

     *   "conference"      - "SUBSCRIBE for conference event package" acc. annex A.5.1; NOTE in this case p_SipUrlTo shall be present and set acc. to "px_FinalConferenceUri"

     *   "message-summary" - "SUBSCRIBE for Message Waiting Indication package" acc. annex A.6.1

     *   "presence"        - "SUBSCRIBE for presence event package MCX TS 36.579-1 Table 5.5.2.14-1

     *   "xcap-diff"       - "SUBSCRIBE for xcap-diff event package, MCX TS 24.379 clause 6.3.5 */

    /* @sic R5-130750 change 4: specific IMPU for messages sent in DL sic@ */

    /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */

    /* @sic R5s150086 - p_SessionIdSupported: pc_IMS_SessionIdSupported per default sic@ */

    /* @sic R5-174544: SessionId removed sic@ */

    var template (present) MessageHeader v_MessageHeader_Subscribe := cr_MessageHeader_Dummy;

    var IMS_ProtectedPorts_Type v_Protected;

    var template (present) PAccessNetworkInfo v_PAccessNetworkInfo := cr_PAccessNetworkInfo(f_IMS_PTC_GetRanType());

    var template (present) SipUrl v_SipUrlFrom := f_IMS_PTC_ImsInfo_GetSipUriSubscriptionRX();

    var template (present) SipUrl v_SipUrlTo;

    var template integer v_Port_ps := *;

    var template integer v_Port_us_Via := *;

    var template integer v_Port_us_Contact := ?;

    var boolean v_CondA2 := f_IMS_PTC_SecurityScheme_IsGiba();       // GIBA

    var boolean v_CondA5 := f_IMS_PTC_SecurityScheme_IsSipDigest();  // FBBA
    var boolean v_IsMCX := f_IMS_PTC_TestSpec_IsMCX();               // MCX 
    var template (present) SemicolonParam_List v_ContactParams := {};
    if (ispresent(p_SipUrlTo)) {

      v_SipUrlTo := p_SipUrlTo;         // NOTE: compiler warning cannot be avoided here

    } else {

      v_SipUrlTo := v_SipUrlFrom;

    }

    v_MessageHeader_Subscribe.cSeq        := cr_CseqDef(?, "SUBSCRIBE");

    v_MessageHeader_Subscribe.fromField   := cr_FromWithTag(v_SipUrlFrom);

    v_MessageHeader_Subscribe.toField     := cr_ToDef(v_SipUrlTo);

    v_MessageHeader_Subscribe.expires     := cr_Expires(p_ExpireTime);  // @sic R5-130548 sic@

    v_MessageHeader_Subscribe.callId      := cr_CallId(?);

    v_MessageHeader_Subscribe.maxForwards := cr_MaxForwardsDef;

    v_MessageHeader_Subscribe.event       := cr_EventDef(p_EventType);

    if (v_CondA2 or v_CondA5) {   // GIBA or FBBA   @sic R5-160290 FBBA registration sic@

      v_MessageHeader_Subscribe.require := omit;

      v_MessageHeader_Subscribe.proxyRequire := omit;

      v_MessageHeader_Subscribe.securityVerify := omit;

      if (v_CondA2) {

        v_MessageHeader_Subscribe.pAccessNetworkInfo := v_PAccessNetworkInfo ifpresent;

      } else {

        v_MessageHeader_Subscribe.pAccessNetworkInfo := v_PAccessNetworkInfo;

      }

    }

    else {                        // IMS security

      v_Protected := f_IMS_PTC_Security_GetProtectedPorts();

      v_Port_ps := v_Protected.Port_ps;

      v_Port_us_Via := v_Protected.Port_us;

      v_Port_us_Contact := v_Protected.Port_us;

      v_MessageHeader_Subscribe.require := cr_RequireSecAgree;

      v_MessageHeader_Subscribe.proxyRequire := cr_ProxyRequireDef;

      v_MessageHeader_Subscribe.securityVerify := f_IMS_PTC_BuildSecurityVerifyHeader();                    // !!!! PROSE ISSUE: Missing in 36.579-1 !!!!
      v_MessageHeader_Subscribe.pAccessNetworkInfo := v_PAccessNetworkInfo

    } 
    if (v_IsMCX) {

      v_MessageHeader_Subscribe.pPreferredService := cr_PPreferredService("urn:urn-7:3gpp-service.ims.icsi.mcptt");

      v_MessageHeader_Subscribe.pAssertedID := cr_PAssertedId(?);                                           // !!!! IMPLEMENTATION ISSUE: shall be px_MCPTT_User_A_ID !!!!

      v_MessageHeader_Subscribe.contentType := cr_ContentType(tsc_IMS_SIP_ContentType_MIME);

      v_ContactParams := f_SemicolonParam_List_Add_RX(v_ContactParams, cr_GenericParam_NoValue("+g.3gpp.mcptt"));

      v_ContactParams := f_SemicolonParam_List_Add_RX(v_ContactParams, cr_GenericParam_NoValue("+g.3gpp.icsi-ref=urn:urn-7:3gpp-service.ims.icsi.mcptt"));

      v_ContactParams := f_SemicolonParam_List_Add_RX(v_ContactParams, cr_FeatureParam_Audio);

    }
    v_MessageHeader_Subscribe.route   := cr_Route(cr_RouteBodyList_InitialReq(v_Port_ps));                  /* @sic R5s140934 change 1 sic@ */

    v_MessageHeader_Subscribe.via     := cr_ViaDef(-, cr_HostPort(?, v_Port_us_Via));                       /* host address needs to be checked after receiving @sic R5s120727 change 6.1 sic@ */

    v_MessageHeader_Subscribe.contact := cr_Contact(f_Contact_SipUri_HostPortRX(?, v_Port_us_Contact), v_ContactParams);     /* host address needs to be checked after receiving @sic R5s120727 change 6.1 sic@

                                                                                                               @sic R5s130925 change 1: UE may use any port as server port sic@

                                                                                                               @sic R5s150721, R5-153760: cr_SipUri_HostPort replaced by f_Contact_SipUri_HostPortRX sic@ */

    select (p_EventType) {

      case ("reg") {                // normal case

        v_MessageHeader_Subscribe.accept := cr_AcceptDef(c_xmlreginfoApplication) ifpresent;

      }

      case ("conference") {         // "SUBSCRIBE for conference event package" acc. annex A.5.1

        v_MessageHeader_Subscribe.accept := cr_AcceptDef("application/conference-info+xml");

        v_MessageHeader_Subscribe.expires := cr_Expires(?);

      }

      case ("message-summary") {    // "SUBSCRIBE for Message Waiting Indication package" acc. annex A.6.1

        v_MessageHeader_Subscribe.accept := cr_AcceptDef("application/simple-message-summary")

      }

      case ("presence") {           // SUBSCRIBE for presence event package MCX TS 36.579-1 Table 5.5.2.14-1

        v_MessageHeader_Subscribe.accept := cr_AcceptDef("application/pidf+xml")

      }

      case ("xcap-diff") {          // SUBSCRIBE for xcap-diff event package, MCX TS 24.379 clause 6.3.5


        v_MessageHeader_Subscribe.accept := cr_AcceptDef("application/pidf+xml");                           // !!!! PROSE ISSUE: should be "application/xcap-diff+xml" !!!!
      }

    }

    return v_MessageHeader_Subscribe;

  }  



3.3 
Altstep a_IMS_SUBSCRIBE_Request
	altstep a_IMS_SUBSCRIBE_Request(out IMS_DATA_REQ p_IMS_DATA_REQ,

                                  EventType p_EventType,

                                  template (present) SipUrl p_SipUrlRequestUri,

                                  template SipUrl p_SipUrlTo := omit,

                                  template (present) charstring p_ExpireTime := "600000",

                                  template MessageBody p_MessageBody := omit) runs on IMS_PTC

  { /* NOTE: applicable only if the SUBSCRIBE creates a new dialog (i.e. is ot sent within existing dialog */

    var IMS_DATA_REQ v_IMS_DATA_REQ_ByRef;

    var MessageHeader v_MessageHeader;

    var InternetProtocol_Type v_Protocol;

    var charstring v_ToTag := f_IMS_GenerateTag("SUBSCRIBE-to-" & p_EventType & "-package");    /* @sic R5-153756: dialog for subscribe sic@ */

    var template (present) MessageHeader v_ExpectedMessageHeader := f_IMS_SubscribeRequest_MessageHeaderRX(p_EventType, p_SipUrlTo, p_ExpireTime);

    var IMS_PTC_DialogIndex_Type v_DialogIndex;

    [] a_IMS_ReceiveRequest(car_IMS_Subscribe_Request(cr_SUBSCRIBE_Request(p_SipUrlRequestUri, v_ExpectedMessageHeader, p_MessageBody)), v_IMS_DATA_REQ_ByRef)

      {

        v_MessageHeader := v_IMS_DATA_REQ_ByRef.Request.Subscribe.msgHeader;

        v_Protocol := v_IMS_DATA_REQ_ByRef.RoutingInfo.Protocol;

        f_IMS_MessageHeader_CheckTo(v_MessageHeader, noToTag);           // This cannot be checked in the template

        f_IMS_MessageHeader_Request_CheckVia(v_MessageHeader, v_Protocol);      // @sic R5s120727 change 9 sic@

        f_IMS_MessageHeader_CheckContactAddr(v_MessageHeader);                  // @sic R5s120727 change 9 sic@

        f_IMS_MessageHeader_Request_CheckRoute(v_MessageHeader);                // @sic R5s140934 change 2 sic@

        if (f_IMS_PTC_ImsInfo_SigCompStarted()) {

          f_IMS_MessageHeader_CheckSigComp(v_MessageHeader);

        }

        select (p_EventType) {

          case ("reg")             { v_DialogIndex := regEventDialog; }

          case ("message-summary") { v_DialogIndex := msgSummaryDialog; }

          case ("conference")      { v_DialogIndex := confEventDialog; }

          case ("presence")        { v_DialogIndex := presenceDialog; }

          case ("xcap-diff")       { v_DialogIndex := xcapDiffDialog; }

          case else                { FatalError(__FILE__, __LINE__, "invalid event type"); }

        }

        f_IMS_PTC_ImsInfo_DialogInit(dialogRemote, v_MessageHeader, v_DialogIndex);               /* @sic R5-153756: dialog for subscribe sic@ */

        f_IMS_PTC_ImsInfo_DialogSetLocalTag(v_ToTag, v_DialogIndex);                              /* @sic R5-153756: dialog for subscribe sic@ */

        p_IMS_DATA_REQ := v_IMS_DATA_REQ_ByRef;

      }

  }


4.
Further changes

4.1
New global variable the distinguish IMS and MCX test cases

	  type enumerated IMS_TestSpec_Type {

    TS_34229_1,

    TS_36579_1

  };

	  type record IMS_Global_Type {                                 /* @status    APPROVED (ENDC, IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, NR5GC, POS) */

    PDN_Index_Type              PdnIndex,                       /* @sic R5s130651: new variable sic@ */

    IMS_PTC_Ctrl_Type           Ctrl,

    IMS_PTC_SecurityInfo_Type   Security,

    IMS_PTC_ImsInfo_Type        ImsInfo ,

    IMS_TestSpec_Type           TestSpec
  };

	  function f_IMS_PTC_TestSpec_IsMCX() runs on IMS_PTC return boolean

  {

    return (vc_IMS_Global.TestSpec == TS_36579_1);

  }


4.2
Change of f_IMS_PTC_Init and f_IMS_CC_Preamble
· p_UE_Release is never handed over to f_IMS_PTC_Init ( can be removed

· New parameter p_DigestPasswordForSIP for f_IMS_PTC_Init to make MCX independent from IMS test case specific PIXIT px_DigestPasswordForSIP
· New parameter p_TestSpec for f_IMS_PTC_Init to initialise new global variable TestSpec 
(see change 3.1)
	  function f_IMS_PTC_Init(PDN_Index_Type p_PdnIndex := PDN_1,

                          IMS_PTC_Handling_Type p_Handling := testcaseSpecificHandling,

                          template (omit) IMS_SecurityScheme_Type p_SecurityScheme := omit,

                          template (omit) IPCAN_RAN_Type p_RanType := omit ,

                          template (omit) charstring p_DigestPasswordForSIP := omit ,

                          IMS_TestSpec_Type p_TestSpec := TS_34229_1
)  runs on IMS_PTC

  { /* @sic R5s130651: new parameter p_PdnIndex sic@ */

    /* @sic R5s130756 additional changes: new parameter p_SecurityScheme sic@ */

    /* @sic R5s130756: new parameter p_Pcscf sic@ */

    /* @sic R5-144747: removal of px_IMS_Pcscf and px_IMS_Pcscf_Emg => p_Pcscf removed sic@ */

    var charstring v_HomeDomainName;

    var charstring v_PrivateUserId;

    var CharStringList_Type v_IMPUList;

    var IMS_SecurityScheme_Type v_SecurityScheme;

    var charstring v_Password_FBBA;

    var boolean v_IsFBBA := match(FBBA, p_RanType);   // @sic R5-160290 FBBA registration sic@

    var IntegerList_Type v_SPI_ValidValueList_PDN1 := fl_IMS_PTC_SPI_ValidValueList_Init(px_IMS_SPI_ValidValueList_PDN1);  /* @sic R5-172048 sic@ */

    var IntegerList_Type v_SPI_ValidValueList_PDN2 := fl_IMS_PTC_SPI_ValidValueList_Init(px_IMS_SPI_ValidValueList_PDN2);  /* @sic R5-172048 sic@ */

    var template (omit) integer v_UE_Release := omit;
    if (px_UEwithISIM or v_IsFBBA) {  /* @sic R5-133628: check of temporary user id sic@ */

      v_HomeDomainName := px_IMS_HomeDomainName;

      v_PrivateUserId := px_IMS_Private_UserId;

    } else {

      v_HomeDomainName := f_DomainName_IMS_DerivedFromIMSI(px_IMSI_Def);

      v_PrivateUserId := f_IMS_PrivateUserIdentityDerivedFromIMSI(px_IMSI_Def);

    }

    if (ispresent(p_SecurityScheme)) {

      v_SecurityScheme := valueof(p_SecurityScheme);

    } else if (pc_IMS_GIBA_Sec and not pc_IMS_Sec) {

      v_SecurityScheme := GIBA;

    } else if (pc_IMS_Sec) {

      v_SecurityScheme := IMS_Security;

    } else {

      FatalError(__FILE__, __LINE__, "Invalid security setting (neither pc_IMS_GIBA_Sec nor pc_IMS_Sec is set)");

    }

    if (v_IsFBBA) {     /* @sic R5-160290 FBBA registration sic@ */

      v_IMPUList := {px_IMS_PublicUserIdentity1};

    } else {

      v_IMPUList := {px_IMS_PublicUserIdentity1, px_IMS_PublicUserIdentity2, px_IMS_PublicUserIdentity3};

    }

    vc_IMS_Global.PdnIndex := p_PdnIndex;

    vc_IMS_Global.Ctrl.TestBody := false;

    vc_IMS_Global.Ctrl.BlockUlDataInDefault := false;       /* @sic R5s140780 - MCC160 Implementation sic@ */

    vc_IMS_Global.Ctrl.IgnoreReset := true;                 /* @sic R5s150031: IMS is already reset after initialisation sic@ */

    vc_IMS_Global.Ctrl.Default := null;

    vc_IMS_Global.Security := valueof(cs_IMS_SecurityInfo_Init);

    if (v_IsFBBA and isvalue(p_DigestPasswordForSIP)) {  /* @sic R5-160914, R5-160915 new TC H.8.1 sic@ */

      /* Authorization and Proxy-Authorization response in Register and other request messages

       * for IMS over Fixed Broadband is MD5 digest based on password (not SIM based AKA)

       */

      v_Password_FBBA := valueof(p_DigestPasswordForSIP; // See 34.299-1 A.1.1, @sic R5-170885 sic@

      vc_IMS_Global.Security.XRES := oct2bit(char2oct(v_Password_FBBA));

      vc_IMS_Global.Security.XRESLength := lengthof(v_Password_FBBA);

    }

    vc_IMS_Global.ImsInfo  := valueof(cs_IMS_ImsInfo_Init(p_RanType, v_IMPUList, v_UE_Release, v_HomeDomainName, v_PrivateUserId, v_SecurityScheme, v_SPI_ValidValueList_PDN1, v_SPI_ValidValueList_PDN2)); 
    vc_IMS_Global.TestSpec := p_TestSpec;
    f_IMS_PTC_ActivateDefault(p_Handling);

  }

	  function f_IMS_CC_Preamble(IMS_TestConfiguration_Type p_TestConfiguration,

                             IMS_InitState_Type p_InitState,

                             template (omit) IMS_SecurityScheme_Type p_SecurityScheme := omit,

                             template (omit) IPCAN_RAN_Type p_RanType := omit,

                             boolean p_SigCompIsActive := false) runs on IMS_PTC

  { /* p_TestProcedure  ..  major test procedure of the test case (=> number/type of DRBs to be configured at SS

     * p_InitState      ..  IDLE => intitial registration including IMS registration is done; RRC connection is released

     *                      NULL => intitial registration is started at the IPCAN but IMS registration shall be controled by the test case */

    /* @sic R5s130133 additional changes: XRES length shall be queried by the IMS PTC i.e. is not returned by f_IMS_IPCAN_StartProcedure_Preamble anymore sic@ */

    /* @sic R5s130347: local GuardTimer sic@ */

    /* @sic R5s130756 additional changes: p_IsGIBA replaced by p_SecurityScheme sic@ */

    /* @sic R5w160208 FBBA: new parameter p_RanType sic@ */

    timer t_GuardTimer := tsc_GuardTimePreamble;

    t_GuardTimer.start;

    f_IMS_PTC_Init(PDN_1, -, p_SecurityScheme, p_RanType, px_DigestPasswordForSIP);      /* local initialisation */

    f_IMS_IPCAN_Init(IPCAN, p_TestConfiguration);               /* initialisation of IPCAN:

                                                                   - RAN type is selected by IPCAN PTC acc. to the PIXITs

                                                                   - cell is created acc. to the RAN type and PIXITs

                                                                   - Test procedure is handed over as for e.g. EUTRA we need to know in advance which DRBs need to be (pre-)configured at the SS

                                                                   p_TestcaseName is present If the test case requires a specific IPCAN initialisation e.g. for cell barring */

    if (p_InitState == IMS_REGISTERED) {

      f_IMS_IPCAN_StartProcedure(IPCAN, IPCAN_InitialRegistration);

      f_IMS_Registration(-, -, p_SigCompIsActive);

      f_IMS_CC_ReleaseConnection(IPCAN_InitialRegistration);

    }

    t_GuardTimer.stop;

  }


